

## Description

`graph twoway lpoly` plots a local polynomial smooth of *yvar* on *xvar*.

## Quick start

Kernel-weighted local polynomial smooth plot of *y* versus *x* with local mean smoothing

```
twoway lpoly y x
```

Same as above, and overlay on a scatterplot to show the observed data

```
twoway scatter y x || lpoly y x
```

Same as above, but with gray markers and a navy blue line

```
twoway scatter y x, mcolor(gray) || lpoly y x, lcolor(navy)
```

Specify the half-width of the kernel to be 110

```
twoway scatter y x || lpoly y x, bwidth(110)
```

Specify a polynomial of degree 3

```
twoway scatter y x || lpoly y x, degree(3)
```

Specify the triangle kernel function

```
twoway scatter y x || lpoly y x, kernel(triangle)
```

Show both the triangle and default Epanechnikov kernel functions

```
twoway scatter y x || lpoly y x, kernel(triangle) || lpoly y x
```

With a separate graph area for each level of categorical variable *catvar*

```
twoway scatter y x || lpoly y x , by(catvar)
```

## Menu

Graphics > Two-way graph (scatter, line, etc.)

## Syntax

```
twoway lpoly yvar xvar [if] [in] [weight] [, options]
```

<i>options</i>	Description
<u>kernel</u> ( <i>kernel</i> )	kernel function; default is <code>kernel(epanechnikov)</code>
<u>bwidth</u> (#)	kernel bandwidth
<u>degree</u> (#)	degree of the polynomial smooth; default is <code>degree(0)</code>
<u>n</u> (#)	obtain the smooth at # points; default is $\min(N, 50)$
<i>cline_options</i>	change look of the line
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

<i>kernel</i>	Description
<code>epanechnikov</code>	Epanechnikov kernel function; the default
<code>epan2</code>	alternative Epanechnikov kernel function
<u>biweight</u>	biweight kernel function
<u>cosine</u>	cosine trace kernel function
<u>gaussian</u>	Gaussian kernel function
<u>parzen</u>	Parzen kernel function
<u>rectangle</u>	rectangle kernel function
<u>triangle</u>	triangle kernel function

`fweights` and `aweights` are allowed; see [\[U\] 11.1.6 weight](#).

## Options

`kernel` (*kernel*) specifies the kernel function for use in calculating the weighted local polynomial estimate. The default is `kernel(epanechnikov)`. See [\[R\] `kdensity`](#) for more information on this option.

`bwidth`(#) specifies the half-width of the kernel, the width of the smoothing window around each point. If `bwidth()` is not specified, a rule-of-thumb bandwidth estimator is calculated and used; see [\[R\] `lpoly`](#).

`degree`(#) specifies the degree of the polynomial to be used in the smoothing. The default is `degree(0)`, meaning local mean smoothing.

`n`(#) specifies the number of points at which the smooth is to be calculated. The default is  $\min(N, 50)$ , where  $N$  is the number of observations.

*cline\_options* specify how the line is rendered and its appearance; see [\[G-3\] \*cline\\_options\*](#).

*axis\_choice\_options* associate the plot with a particular  $y$  or  $x$  axis on the graph; see [\[G-3\] \*axis\\_choice\\_options\*](#).

*twoway\_options* are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [\[G-3\] \*twoway\\_options\*](#).

## Remarks and examples

`graph twoway lpoly yvar xvar` uses the `lpoly` command—see [R] [lpoly](#)—to obtain a local polynomial smooth of *yvar* on *xvar* and uses `graph twoway line` to plot the result.

Remarks are presented under the following headings:

*Typical use*


*Use with `by()`*

### Typical use

The local polynomial smooth is often graphed on top of the data, possibly with other smoothers or regression lines:

```
. use https://www.stata-press.com/data/r19/auto
(1978 automobile data)

. twoway scatter weight length      ||
      lpoly weight length          ||
      lowess weight length
```




## Use with by()

graph twoway lpoly may be used with by():

```
. use https://www.stata-press.com/data/r19/auto, clear
(1978 automobile data)

. twoway scatter weight length      ||
      lpoly weight length
      , by(foreign)
```



## References

- Cox, N. J. 2005. [Speaking Stata: Smoothing in various directions](#). *Stata Journal* 5: 574–593.
- . 2010. [Software Updates: Speaking Stata: Smoothing in various directions](#). *Stata Journal* 10: 164.

## Also see

- [R] [lpoly](#) — Kernel-weighted local polynomial smoothing
- [G-2] [graph twoway lpoly](#) — Local polynomial smooth plots with CIs

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

